
Wstęp


Znaczny rozwój języków wysokiego poziomu nie zahamował rozwoju i znaczenia języków symbolicznych (asemblerów). W wielu zastosowaniach, szczególnie tam, gdzie potrzebny jest minimalny kod programu oraz duża szybkość działania programu asembler jest niezastąpiony.


W wielu systemach sterowania, nawet jeżeli programy użytkowe są napisane w językach wysokiego poziomu, niektóre procedury (np. związane z obsługą dużej liczby przerwań przychodzących z obiektu) są tworzone w języku asemblera w celu zapewnienia odpowiednich reżimów czasowych, narzuconych przez proces technologiczny lub proces sterowania.





Asembler


Każdy mikroprocesor realizuje rozkazy, które dostarczone mu są w postaci binarnej (ciągi zer i jedynek). Jest to tzw. język wewnętrzny procesora. Programowanie bezpośrednio w tym języku jest niezwykle uciążliwe i podatne na błędy. Opracowano więc specjalne programy tłumaczące zwane asemblerami (ang. assemble - składać), których zadaniem jest tłumaczenie programów źródłowych w języku symbolicznym (assemblerowym) na język wewnętrzny mikroprocesora. 


Język assemblerowy pozwala na użycie kodów symbolicznych rozkazów mikroprocesora (tzw. mnemoników), zapewnia możliwość adresowania symbolicznego i względnego, zapisywanie argumentów rozkazów w dogodnej postaci oraz w zależności od złożoności assemblera (assemblery proste, makroassemblery) stosowanie makroinstrukcji i mechanizmów zbliżonych do języków wysokiego poziomu. 


Ze względu na różnice wynikające z budowy, rodzaju i ilości realizowanych rozkazów występują różnice w językach symbolicznych dla różnych mikroprocesorów. Znając jednak ogólne zasady programowania w asemblerze, nie trudno jest się "przestawić" na język dla innego miroprocesora. Ponadto warto również nadmienić, że w ramach jednej rodziny mikroprocesorów lista rozkazów procesora wcześniejszego jest przważnie podzbiorem listy rozkazów procesora nowszego lub jest do niej bardzo zbliżona. Tak właśnie jest w przypadku rodziny procesorów INTEL 8088/86/186/286/386/486.


Poniższe opracowanie opiera się na asemblerze tych właśnie procesorów oraz mikrokomputera jednoukładowego INTEL 8051.





Ogólne zasady programowania w asemblerze


Programowanie w asemblerze polega na przygotowaniu zbiorów źródłowych w języku symbolicznym. Język symboliczny składa się z instrukcji (mnemoników), dyrektyw, nazw i adresów symbolicznych (zmienne, etykiety), symboli (rejestry), stałych i komentarzy. Asembler tłumaczy (asemblacja) te zbiory na język wewnętrzny mikroprocesora.





Instrukcje


Instrukcje (rozkazy, mnemoniki) są nierozerwalnie związane z typem procesora. Istnieją  jednak pewne grupy instrukcji charakterystyczne dla wszystkich procesorów. 


Są to:





Rozkazy transmisji danych


przykłady


MOV	przesłanie bajtu lub słowa


PUSH	przesłanie słowa na szczyt stosu


POP		pobieranie słowa ze szczytu stosu


IN		wprowadzenie bajtu lub słowa z układu wejściowego


OUT		wyprowadzenie bajtu lub słowa do układu wyjściowego





Rozkazy operacji arytmetycznych i logicznych


przykłady


ADD		dodawanie bajtów lub słów


INC		inkrementacja


SUB		odejmowanie bajtu lub słowa


DEC		dekrementacja


CMP		porównanie bajtów lub słów


MUL		mnożenie bajtów lub słów


AND 	iloczyn logiczny bajtów lub słów


OR		suma logiczna bajtów lub słów


XOR		suma modulo 2 bajtów lub słów


SHL		przesunięcie logiczne w lewo bajtu lub słowa


SHR		przesunięcie logiczne w prawo bajtu lub słowa





Rozkazy sterowania skokami


przykłady


JMP		zwykły skok bezwarunkowy


CALL	wywołanie podprogramu


RET		powrót z podprogramu do miejsca wywołania


JNE		skok gdy nie są równe


JE		skok gdy są równe


JZ		skok gdy liczba równa zero





Rozkazy przerwań


przykłady


INT		wywołanie podprogramu obsługi przerwania


IRET	powrót z podprogramu obsługi przerwania





Rozkazy sterujące procesorem


przykłady


STD		ustawienie znacznika kierunku (określa zachowanie rej. indeksowych w trakcie wykonywania operacji łańcuchowych)


CLD		wyzerowanie znacznika kierunku


STI		ustawienie znacznika przerwania


CLI		wyzerowanie znacznika przerwania 





Dyrektywy


Dyrektywy mają w języku asemblera specjalne znaczenie. Są one nazwami zastrzeżonymi, co oznacza, że nie mogą być one użyte do innych celów (zmienne, etykiety).


Przykłady dyrektyw


DB, DW	deklaracja bajtu, słowa


ORG		wskaźnik pozycji asemblera w czasie asemblacji


SEGMENT	definje segment


PROC	definuje początek procedury


ENDP	definuje koniec procedury


EQU		pozwala przypisać nazwie symbolicznej określoną wartość


END		wskazuje koniec programu źródłowego





Powyższa lista rozkazów i dyrektyw jest oparta o mikroprocesor 8086. Ze względu na charakter tego opracowania nie jest to pełna lista dla tego procesora. Zainteresowanych odsyłam do poz. I z wykazu literarury.





Nazwy symboliczne


Nazwy symboliczne są to ciągi znaków, które mają dla asemblera szczególne znaczenie. Służą one do identyfikacji zmiennych, stałych, etykiet (adresów symbolicznych), segmentów i nazw procedur. Pierwszy znak nazwy musi być literą lub jest jednym ze znaków specjalnych.





Symbole


Symbolami mogą być w szczególności nazwy rejestrów. Nazwy te zależą od konkretnej architektury procesora. Kilka ogólnych zasad w nazewnictwie rejestrów:





rejestr akumulatora oznaczany jest jako A lub AX


nazwa młodszej części rejestru zakończona jest literą 'L', starsza literą 'H'.





Stałe


Stałe to wartości, które muszą być znane podczas tłumaczenia (asemblacji) programu źródłowego i nie mogą być zmieniane w czasie wykonywania programu. Wyróżnia się stałe numeryczne, będące liczbami całkowitymi oraz stałe łańcuchowe.


Stałe numeryczne mogą być zapisane w systemie dwójkowym, ósemkowym, dziesiętnym lub szesnastkowym.


przykłady


stała dwójkowa	010101010B, 11111111b


stała ósemkowa	34325Q, -345Q


stała dziesiętna	-1994, 8, 124D


stała szesnastkowa	2H, -89h, 0FFH





Tryby adresacji


Rozkazy procesorów mogą być bezargumentowe, mogą też zawierać jeden lub kilka argumentów. Argumentami mogą być: rejestry, argumenty bezpośrednie lub argumenty w pamięci operacyjnej. W zależności od typów argumentów rozróżnia się następujące tryby adresacji:





adresowanie rejestrowe - argument znajduje się w rejestrze, np.


	ADD AX, BX		;dodaj do zawartości akumulatora AX zawartość rej. BX


adresowanie bezpośrednie - adres efektywny zawarty jest bezpośrednio w rozkazie


	ADD AX, 0B9H	;dodanie do zawartości akumulatora wartości 0B9H


adresowanie pośrednie rejestrowe - adres efektywny znajduje się w rejestrze, np.


	MOV AX,[BX]	;prześlij do akumulatora zawartość komórki pamięci o 				offsecie określonym w rej. BX


adresowanie indeksowe - adres efektywny jest sumą adresu bazowego i rejestru indeksowego, np.


	MOV AX, [BP+SI]	;prześlij do akumulatora zawartość komórki o offsecie 				określonym przez sumę zawartości rejestru BP i SI


adresowanie pośrednie (rejestrowe i indeksowe) z przemieszczeniem - jak wyżej, plus przemieszczenie, np.


	MOV AX, [BP+SI+2]


	MOV [DI-2], DX





Asembler mikroprocesora 8088/86





Architektura mikroprocesora 


Aby zrozumieć zasady programowania w asemblerze mikroprocesora 8086 konieczne jest zapoznanie się z jego architekturą wewnętrzną.


Na rysunku 1. przedstawiono rejestry mikroprocesorów 8088/86 dostępne programowo. Są one 16-to bitowe i można wśród nich wyróżnić:





Rejestry ogólnego przeznaczenia


Rejestry te są wykorzystywane do przechowywania dowolnych danych i wykonywania różnych operacji (np.arytmetycznych bądź logicznych), ale jednocześnie pełnią pewne funkcje specjalne odpowiadające ich nazwom


AX (ang. accumulator)	akumulator


BX (ang. basis register)	rejestr bazowy


CX (ang. counter register)	rejestr zliczający


DX (ang. data register)	rejestr danych


Wszystkie rejestry można też traktować jako pary rejestrów 8-bitowych z niezależnym dostępem do części mniej (ang. low) oraz bardziej (ang. high) znaczącej. Poszczególne rejestry przyjmują nazwy: AL, AH, BL, BH, CL, CH, DL, DH. Taka organizacja rejestrów pozwala na wykonywanie w optymalny sposób operacji 8-bitowych, jak i 16-bitowych. Dla wielu rozkazów szczególną funkcję pełni akumulator AX (lub AL). Rozkazy wykorzystujące rejestr AX (w przeciwieństwie do rozkazów wykorzystujących pozostałe rejestry) mają zwarty kod, a czas ich wykonania jest najkrótszy.





Rejestry wskaźnikowe i indeksowe


Mikroprocesory 8088/86 posiadają dwa rejestry wskaźnikowe i dwa indeksowe. Wszystkie są rejestrami 16-bitowymi.


SP (ang. stack pointer)		wskaźnik stosu


BP (ang. base pointer)		wskaźnik bazy


SI (ang. source index register)	rejestr indeksu źródła


DI (ang. destination index register)	rejestr indeksu przeznaczenia





Oba rejestry wskaźnikowe są stosowane do adresowania danych w obrębie wydzielonego obszaru pamięci zwanego segmentem stosu (bliższe wyjaśnienie zagadnień związanych z segmentami znaleźć można w pozycji I ze spisu literatury).


Rejestr SP jest stosowany przy standardowych operacjach zapisu i odczytu stosu, rejestr BP natomiast podczas operacji niestandardowych, np. przy pobieraniu parametrów przekazywanych przez stos. Z kolei oba rejestry indeksowe służą do adresowania danych w obszarze pamięci zwanym segmentem danych. Wszystkie cztery rejestry określają 16-bitowy adres względem początku danego segmentu (ang. offset). Ponadto mogą być one wykorzystane jako argumenty większości rozkazów arytmetycznych i logicznych.





Rejestry segmentowe


Rejestry te są wykorzystywane do adresowania pamięci operacyjnej. Pamięć o maksymalnej wielkości 1 MB jest podzielona na logiczne segmenty nie większe niż 64 KB. Tak więc adresy wewnątrz segmentu są co najwyżej dwubajtowe i mogą być generowane przez 16-bitowe rejestry np. SI, DI, BP, SP.


Rejestry segmentowe zawierają adresy początkowe segmentów.


CS (ang. code segment register)	rejestr segmentowy programu wskazuje segment programu, z którego aktualnie są pobierane kolejne rozkazy do wykonania;


DS (ang. data segment register)	rejestr segmentowy danych wskazuje segment, w którym są zapamiętane zmienne używane w programie;


ES (ang. extra segment register)	rejestr segmentowy dodatkowy wskazuje dodatkowy segment danych


SS (ang. stack segment register)	rejestr segmentowy stosu wskazuje segment pamięci, w którym jest zdefiniowany stos.





Wskaźnik rozkazów


Rejestr IP zwany wskaźnikiem rozkazów (ang. instruction pointer), łącznie z rejestrem segmentowym CS adresuje kolejne rozkazy przeznaczone do wykonania. 





Rejestr znaczników


Rejestr FLAGS (ang. flag register) zawiera zestaw znaczników przypisanych poszczególnym bitom. Znaczniki te można podzielić na dwie grupy:





Znaczniki stanu - informujące o pewnych cechach otrzymanego wyniku po wykonaniu operacji arytmetycznej bądź logicznej. W zależności od stanu pojedynczych znaczników lub ich logicznej kombinacji można za pomocą włściwych rozkazów skoków zmienić przebieg realizowanego programu





CF (ang. carry flag)		przeniesienie


PF (ang. parity flag)		parzystość


AF (ang. auxiliary flag)	przeniesienie pomocnicze


ZF (ang. zero flag)		zero


SF (ang. sign flag)		znak


OF (ang. overflow)		nadmiar (przepełnienie)





Znaczniki kontrolne	ustawiane lub zerowane programowo celem wymuszania odpowiedniego sposobu pracy procesora.


TF (ang. trap flag)			praca krokowa (pułapka)


IF (ang. interupt enable flag)	zezwolenie na przerwanie


DF (ang. direction flag)	znacznik kierunku.

Praca z asemblerem


Najbardziej popularnymi asemblerami dla mikroprocesorów 8088/86 pracującymi w systemie MS DOS są programy MASM (MS Assembler) firmy Microsoft oraz TASM (Turbo Assembler) firmy Borland International. Zasady pracy z tymi programami są w zasadzie identyczne. W poniższym opracowaniu oparto się na asemblerze firmy Borland International.


Asembler ten składa się z dwóch programów:


tasm	program tłumaczący (właściwy asembler)


tlink	program konsolidujący





Uruchomienie programu w asemblerze polaga na przygotowaniu zbioru źródłowego w dowolym edytorze tekstowym zapisującym w standardzie ASCII (edit z MS DOS'a, MultiEdit, Norton Editor itp.). Następnie zbiór taki powinien zostać poddany asemblacji. 








W przypadku asemblera TASM należy wywołać:





tasm test	lub


tasm test.asm		(jeżeli nasz zbiór źródłowy ma nazwę 'test')





W wyniku tej operacji powstał zbiór test.obj, który jest przejściowym zbiorem binarnym (nie można go uruchomić). Zbiór ten należy poddać konsolidacji przy pomocy programu tlink. 





tlink test  (lub tlink test.obj)





W wyniku konsolidacji powstał program test.exe gotowy do uruchomienia. 	Oczywiście zgodnie z prawami Murphy’ego program ten nie będzie wolny od błędów, dlatego w następnym kroku należy przystąpić do procesu uruchamiania stworzonego programu zwanego debugowaniem (z ang. odpluskwianie).


Proces lokalizacji i usuwania błędów może być żmudny i czasochłonny. Bardzo często zdarza się, że programista poświęca znacznie więcej czasu na lokalizację i usunięcie błędu niż na napisanie programu. 





Proces usuwania błędów można najogólniej podzielić na cztery kroki: 





	1. Ustalenie, że występuje błąd 


	2. Znalezienie błędu 


	3. Ustalenie powodu wystąpienia błędu 


	4. Usunięcie błędu 





Czy jest błąd ? 


Pierwszy z kroków może być zupełnie oczywisty. Program powoduje zawieszenie systemu lub lub ekran pokrywa się "sieczką" znaków. Czasem jednak występowanie błędu nie jest tak oczywiste. Program pracuje poprawnie dopóki wprowadzimy nietypowe dane wejściowe lub liczba danych jest duża. 





Gdzie znajduje się  błąd ? 


Krok drugi (odnalezienie błędu) może być czasem najtrudniejszy. Najprostszą metodą będzie tutaj zastosowanie reguły "Dziel i zwyciężaj". Program należy podzielić na mniejsze części i testować je osobno. Programowanie strukturalne znacznie ułatwia taki proces. 





Co to za błąd ? 


Krok trzeci - ustalenie powodu błędu jest drugą, najtrudniejszą czynnością w procesie "odpluskwiania".  Kiedy ustali się, gdzie wystąpił błąd, to dokładne jego wykrycie i rozpoznanie będzie już ułatwione. Na przykład, jeżeli wykryto, że błąd występuje w procedurze PiszImiona, to wystarczy sprawdzić linie tej procedury. Nie wyklucza to jednak faktu, że w dalszym ciągu błąd może być  złożony i będzie trzeba trochę poeksperymentować zanim zostanie wyeliminowany. 





Usunięcie 


Ostatnim krokiem jest usunięcie błędu. Będąc uzbrojonym w wiedzę o strukturze i zasadach języka programowania można "zgnieść" pluskwę. Następnie należy ponownie uruchomić program , znaleźć następny błąd i rozpocząć proces debuggingu od nowa. 


Nie wszystkie błędy można wyizolować i usunąć w wyżej wymieniony sposób bez dodatkowej pomocy. Wiele z błędów czyha w ukryciu i czeka na najbardziej niespodziewany moment aby uderzyć.... Bez pomocy Debuggera niejednokrotni nie zdoła się zapobiec takiemu uderzeniu. 





Debugger


Debugger jest programem-narzędziem ułatwiającym programiście proces usuwania błędów. Jeżeli umożliwia on "debugging" na poziomie źródłowym programu (dostępne symboliczne nazwy zmiennych i etykiet) to debugger taki będzie nazywany symbolicznym. W środowisku systemu DOS istnieje kilka popularnych debuggerów. Najprostszym debuggerem jest program Debug znajdujący się standardowo w systemie. Nie jest to debugger symboliczny. Najbardziej popularnymi debuggerami symbolicznymi są program Turbo Debugger (td) firmy Borland International oraz Code View (cv) firmy Microsoft. Działanie tych debuggerów jest podobne. Umożliwiają one powolne wykonanie programu, i pozwalają na analizę stanu programu w dowolnym punkcie. 





Najważniejsze cechy nowoczesnych debuggerów symbolicznych: 





Śledzenie (Tracing) - umożliwia wykonywanie programu linia po lini 


Śledzenie wstecz (Back tracing) - umożliwia wykonywanie programu wstecz (powrót do ostatniej instrukcji) 


Przeskakiwanie (Stepping) - umożliwia wykonywanie programu linia po lini z możliwością przeskakiwania wywołań funkcji lub procedur. Jeżeli jest się pewnym, że procedura lub funkcja jest bezbłędna to przyspiesza to proces debuggingu. 


Przeglądanie (Viewing) - umożliwia podglądanie aktualnego stanu programu (zmienne, rejestry, stos, pamięć, pliki, ekran) 


Inspekcja (Inspecting) - podglądanie stanu bardziej złożonych zmiennych (rekordy, tablice) 


Modyfikacja (Changing) - zmiana aktualnej wartości zmiennej na wartość zadaną 


Obserwacja  (Watching) - umożliwia podgląd wybranych zmiennych 





Oprócz usuwania błędów debugger może być zastosowany do tropienia działania "zarażonych" wirusem plików oraz podglądania działania innych programów lub systemu. 








Turbo Debugger 


W ramach laboratorium używany będzie debugger firmy Borland (Turbo Debugger 2.5) wchodzący w skład pakietu Borland C++ 2.0. Program ten jest debuggerem symbolicznym umożliwiającym uruchamianie i obserwację programów źródłowych stworzonych w językach Turbo (Turbo Pascal ,Turbo (Borland) C, C++, Turbo Assembler). Program posiada przyjazny dla użytkownika interfejs. Dostępne jest wielopoziomowe menu, informacje wyświetlane są w okienkach, dane wprowadza się w oknach dialogowych. Turbo debugger zapewnia wszystkie wymienione w poprzednim punkcie cechy nowoczesnego debuggera. 





�


Rys 1. Wygląd ekranu podczas pracy z Turbo Debuggerem  (debuggowany program w postaci symbolicznej)





�


Rys 2. Wygląd ekranu podczas pracy z Turbo Debuggerem  (okno CPU)





Opcje 


Opcje Turbo Debuggera mogą być wybrane bezpośrednio z menu lub przy pomocy klawiszy skrótu. Menu programu podzielone jest na osiem pozycji. Każda z pozycji posiada podmenu. Zasady wyboru pozycji menu, wyświetlania okien, przełączania się pomiędzy nimi są tutaj zgodne z ogólnymi zasadami obowiązującymi w środowisku produktów firmy Borland. Poniżej przedstawione zostaną najważniejsze opcje programu Turbo Debugger. Przedstawiono je w kolejności odpowiadającej występowaniu ich w menu.


 


File 


Open 		otwiera program do debuggingu. Program ten musi być 	typu EXE lub COM. 


Get Info 		wyświetla informacje o programie. 


Resident 		powoduje że Turbo Debugger przechodzi do stanu 	rezydentnego 


Quit 		powrót do DOS'a 





View 


Breakpoints 		pokazuje punkty w których nastąpi zatrzymanie 	programu 


Stack 		pokazuje aktualny stan stosu programu 


Log 		pokazuje okno z raportem 


Watches 		pokazuje zmienne, które poddane są obserwacji 


Variables 		pokazuje globalne i lokalne  zmienne 


Module 		pokazuje odpowiedni moduł programu 


File 		pokazuje zbiór jako ASCII lub HEX 


CPU 		pokazuje aktualny stan procesora 


Dump 		pokazuje zawartość pamięci 


Registers 		pokazuje okno rejestrów 


Another 		tworzy dodatkowe okno z zawartością pamięci, 	modułem lub plikiem 





Run 


Run 		uruchamia załadowany program bez zatrzymania (F9) 


GoToCursor 		uruchamia program aż do aktualnej pozycji kursora (F4) 


Trace Into 		uruchamia pojedyńczą linię programu lub instrukcję (F7) 


Step Over 		uruchamia pojedynczą linię programu nie wchodząc do 	wywoływanych procedur (F8) 


Execute To 		uruchamia program aż do wybranego adresu  (Alt F9) 


Until Return		uruchamia program do momentu wyjścia z funkcji 	(AltF8) 


Animate 		uruchomienie programu "w zwolnionym tempie" 


Back Trace 		powoduje wykonanie programu wstecz o jedną linię lub 	i nstrukcję (Alt F4) 


Arguments 		ustawia argumenty programu (command line) 


Program Reset 		przeładowuje aktualny program i ustawia stan 	początkowy (Ctrl F2) 





Breakpoints 


Toggle 		włącza/wyłącza breakpoint w lini z kursorem  (F2) 


At 		ustawia breakpiont pod wprowadzonym adresem 	(AltF2) 


Changed Memory Global	ustawia breakpoint na zmianę wybranego obszaru 	pamięci 


Expresion True Global 		ustawia breakpoint na wyrażenie 


Delete All 		usuwa wszystkie breakpoint-y 





Data 


Inspect 		podgląd stanu obiektu 


Evaluate/Modify 		wartościuje wyrażenie (Ctrl F4) 


Add Watch		dodanie zmiennej do obserwacji (Ctrl F7) 


Function Return 		sprawdza wartość zwracaną przez aktualną funkcję 





Options 


Language 		ustala język wyrażeń dla modułu 


Macro    		definiowanie makroinstrukcji 


Display  		pokazuje opcje konfiguracji 


Save Options		zapisuje opcje konfiguracji


Restore Options 		odtwarza opcje z dysku 





Window 


Help 


Opcje standardowe dla języków rodziny Turbo 








Uruchomienie Turbo Debuggera 


Aby uruchomić Turbo Debugger należy napisać w lini zleceń: td lub td nazwa_programu, który będzie poddany procesowi wyszukiwania błędów. Aby skorzystać z możliwości "debuggingu" symbolicznego należy zapewnić by program uruchamiany posiadał informację symboliczną (symbol table). W przypadku programu napisanego w Turbo Asemblerze należy użyć przełącznika /zi przy asemblacji oraz przełącznika /v podczas linkowania. 





tasm /zi program.asm


tlink /v program.obj








Przykłady	<


Poniżej przedstawiono przykładowe programy w asemblerze 8088/86. 





Program I





;**********************************************************


;* Program kopiujący blok bajtów od adresu SOURCE_A


;* zakończonych znakiem '#', na ekran od offsestu 800. 


;* Po każdym przekopiowanym znaku nastepuje 


;* oczekiwanie na przyciśnięcie klawisza.


;**********************************************************





KONIEC EQU 35		;znak '#' - zakończenie bloku bajtów


DEST   EQU 0B800H	;segment pamięci ekranu karty 						;EGA,VGA w trybie tekstowym





;*******************************************************


CODE    SEGMENT


        ASSUME CS:CODE,DS:CODE


        ORG 100H             ; Program typu .COM, CS=DS








START:  MOV  AX,CS            


        MOV  DS,AX           ; DS <- CS





        MOV  SI,0		; Inicjacja indeksu źrźdła


        MOV  DI,800		; Inicjacja indeksu


					; przeznaczenia   (5*160)





        MOV  AX,DEST 		; AX <- Segment pamięci ekranu


        MOV  ES,AX 		; Inicjuj pomocniczy rejestr 


					; segmentowy








LOAD:   MOV  AL,SOURCE_A[SI]	; AL <- SOURCE_A[i]


        CMP  AL,KONIEC      	; czy koniec danych ?


        JE   FINITO         	; tak, więc kończymy


        MOV  AH,07H        	; atrybut do starszego 


						; bajtu


        CALL WAIT_KEY       	; Czekaj na klawisz


        MOV  ES:[DI],AX    	; DEST[800 + i] <- AX


        INC  SI            	; SI <- SI + 1


        ADD  DI,2         	; DI <- DI + 2 ponieważ 


					; pamięć ekranu ma 


					; organizację: młodszy 


					; bajt: znak, starszy: 


					; atrybut


        JMP  LOAD          	; skok pod etykietę LOAD





FINITO: MOV AX,4C00H      	; powrót do DOS'a


        INT 21H








;*******************************************************


;* Procedura oczekiwania na przyciśnięcie klawisza


;*******************************************************


WAIT_KEY PROC NEAR


       PUSHF                  ;zachowaj rejestry


       PUSH AX


       MOV AH,00H


       INT 16H


       POP AX


       POPF                   ;odtwórz rejestry


       RET


ENDP


;*******************  Dane  ***********************


SOURCE_A DB 'Studium Podyplomowe. Laboratorium z Assemblera'


,KONIEC





CODE ENDS


END START








Program 2








;******************************************************************************


;* Program kopujacy blok bajtow od adresu SOURCE_A zakonczonych znakiem '#'


;* na ekran od offsestu 800. Po kazdym przekopiowanym znaku nastepuje oczekiwanie


;* na przycisniecie klawisza.


;* ponadto program modyfikuje zmienna index (inkrementacja zmiennej w


;* petli glownej oraz przepisuje zawarosc tablicy SOURCE_A do tablicy


;* DEST_A w odwrotnej kolejnosc


;******************************************************************************





KONIEC 		EQU 35		;znak '#' - zakonczenie bloku bajtow


DEST   		EQU 0B800H	;segment pamieci ekranu karty EGA,VGA w trybie 						;tekstowym


ASIZE  		EQU 50





;******************************************************************************


CODE SEGMENT


		ASSUME CS:CODE,DS:CODE


		ORG 100H    		; Program typu .COM








START:	MOV  AX,CS            


		MOV  DS,AX            	; DS <- CS





		MOV  SI,0             	; Inicjacja indeksu zrodla


		MOV  DI,830           	; Inicjacja indeksu przeznaczenia


		MOV  AX,DEST


		MOV  ES,AX            	; Inicjuj pomocniczy rejestr segmentowy





		CALL  CLS              	; czysc ekran


		MOV   INDEX,ASIZE





LOAD:	MOV  AL,SOURCE_A[SI]  	; AL <- SOURCE_A[i]


        		CMP  AL,KONIEC        		; czy koniec danych ?


        		JE   FINITO           		; tak, 	wiec konczymy


        		MOV  AH,07H           		; laduj atrybut do starszego bajtu


        		CALL WAIT_KEY         		; Czekaj na klawisz


        		MOV  ES:[DI],AX       		; DEST[800 + i] <- AX


        		MOV  DX,DI            		; zachowaj DI


        		MOV  DI,INDEX         		; di indeksuje teraz tablice DEST_A


        		MOV  DEST_A[DI],AL    		; wyslij do tablicy


        		MOV  DI,DX            		; odtworz DI


        		INC  SI               		; SI <- SI + 1


        		ADD  DI,2             		; DI <- DI + 2 poniewaz pamiec ekranu ma 						; organizacje:


                           		   		; mlodszy bajt - znak, starszy - atrybut


        		DEC  INDEX


        		JMP  LOAD             		; skok pod etykiete LOAD





FINITO: 	MOV AX,4C00H          		; powrot do DOS'a


        		INT 21H


;*****************************************************************************


;* Procedura oczekiwania na przycisniecie klawisza


;*****************************************************************************


WAIT_KEY PROC NEAR


       	PUSHF


       	PUSH AX


       	MOV AH,00H


       	INT 16H


       	POP AX


       	POPF


       	RET


ENDP








;*****************************************************************************


;* Czyszczenie ekranu


;*****************************************************************************


CLS PROC NEAR


        	PUSH DI


        	MOV DI,0


        	MOV AX,0720H       ;SPACJA


        	MOV CX,2000


        	CLD


CLS1:	STOSW              ; AX -> [ES:DI]


        	LOOP CLS1


        	POP  DI


        	RET


ENDP





;******************************  Dane  **************************************


SOURCE_A DB 'Studium Podyplomowe. Laboratorium:  Debuggery', KONIEC


DEST_A   DB ASIZE DUP (32)


INDEX    DW 0


CODE ENDS


END START








Program 3





;****************************************************************************


;* Odpluskwic !!!


;****************************************************************************





KONIEC 	EQU 35     	   	;znak '#' - zakonczenie bloku bajtow


DEST   	EQU 0B800H  	;segment pamieci ekranu karty EGA,VGA w 						;trybie tekstowym


ASIZE  	EQU 50





;******************************************************************************


CODE 	SEGMENT


        		ASSUME CS:CODE,DS:CODE


        		ORG 100H 		             ; Program typu .COM





 


START:  	MOV  AX,CS            


       		MOV  DS,AX            








        		MOV  SI,0             


        		MOV  DI,830           





        		MOV  AX,DEST


        		MOV  ES,AX            





        		CALL  CLS          








LOAD:   	MOV  AL,SOURCE_A[SI]


        		CMP  AL,KONIEC


        		JE   LOAD1


        		MOV  AH,07H


        		MOV  ES:[DI],AX





        		INC  SI


        		ADD  DI,1





        		JMP  LOAD





LOAD1:  	MOV  SI,0





        		MOV  AL,SOURCE_A[SI]


        		MOV  SOURCE_A[DI],AH


        		INC  SI


        		DEC  DI


        		DEC  DI


        		JMP  LOAD1





FINITO: 	MOV AX,4C00H          ; powrot do DOS'a


        		INT 21H

















;*****************************************************************************


;* Czyszczenie ekranu


;*****************************************************************************


CLS PROC NEAR


	PUSH DI


       	MOV DI,0


        	MOV AX,0720H    


        	MOV CX,2000


        	CLD


  CLS1:STOSW           


       	LOOP CLS1


        	POP  DI


        	NOP ;DI


        	RET


ENDP











;******************************  Dane  **************************************


SOURCE_A DB 'Cos jest nie tak jak ma byc';KONIEC


DEST_A   DB ASIZE DUP (32)


INDEX    DW 0


CODE ENDS


END START

















Program 4





;*************************************************************


;* Program testujący przyciski myszki. Po naciśnięciu lewego 


;* przycisku "zapala się" odpowiedni punkt na ekranie. 


;* Gaszenie punktu następuje po naciśnięciu dowolnego klawisza


;* klawiatury. Wyjście z programu natępuje przez przyciśnięcie


;* środkowego przycisku myszki. 


;*************************************************************





DEST   EQU 0B800H	;segment pamieci ekranu 


				;karty EGA,VGA w trybie tekstowym





;***********************************************************


CODE    SEGMENT


        ASSUME CS:CODE,DS:CODE


        ORG 100H  		; Program typu .COM





 


START:  MOV  AX,CS      


        MOV  DS,AX   		; DS <- CS;  CS = DS





        MOV  AX,DEST  		; AX <- Segment pamięci ekranu


        MOV  ES,AX    		; Inicjuj pomocniczy 


					; rejestr segmentowy








        CALL RESET_MOUSE 	; Inicjacja myszy


        CALL SHOW_MOUSE  	; Pokaż kursor





QUERY_MOUSE:


        MOV  AX,03H  		; - odczytaj parametry myszy


        INT  33H    		; /


        MOV  AX,BX     	; zachowaj BX


        AND  AX,0001H        ; czy naciśnięto lewy ?


        JNZ  LEFT_BUTTON     ; tak więc obsłuż


        MOV  AX,BX


        AND  AX,0002H   	; czy naciśnięto prawy ?


        JNZ  RIGHT_BUTTON  	; tak więc obsłuż


        MOV  AX,BX


        AND  AX,0004H      	; czy naciśnięto środkowy ?


        JNZ  FINITO      	; tak, kończymy program


        JMP  QUERY_MOUSE 	; zawracamy








FINITO: CALL HIDE_MOUSE   	; zrób porządek z myszą


        MOV AX,4C00H         ; powrót do DOS'a


        INT 21H








; **** obsługa naciśnięcia lewego przycisku





LEFT_BUTTON: MOV AL,'L'  	; parametr dla procedury INDICATE


             CALL INDICATE


             JMP QUERY_MOUSE

















; **** obsługa naciśnięcia prawego przycisku





RIGHT_BUTTON: MOV AL,'R' 	; parametr dla procedury INDICATE


              CALL INDICATE


              JMP QUERY_MOUSE








;**********************************************


;* Procedura realizująca oczekiwanie na klawisz


;**********************************************


WAIT_KEY PROC NEAR


       PUSHF


       PUSH AX


       MOV AH,00H


       INT 16H


       POP AX


       POPF


       RET


ENDP





;********************************************************


;* Procedura wyświetlająca w zależności od naciśniętego 


;* przycisku podswietloną literę L lub R. Po wyświetleniu 


;* użytkownik naciska dowolny klawisz, litera zniknie.


;* Argumenty: AL - 'R' - wyświetl prawy


;*                 'L' - wyświetl lewy


;********************************************************


INDICATE PROC NEAR


       MOV AH,70H      	; załaduj atrybut (podświetlenie)


       CMP AL,'R'    	 	; czy prawy ?


       JE I_R         		; tak, więc skocz do wyświetlania 


					; dla prawego


       MOV ES:[860],AX 	; wyświetl dla lewego


       JMP I_E        		; skocz do oczekiwania na klawisz


  I_R: MOV ES:[900],AX 	; wyświetl dla prawego


  I_E: CALL WAIT_KEY   	; oczekiwanie na klawisz


       MOV ES:[860],0720H 	; wymazanie z ekranu (znak spacji)


       MOV ES:[900],0720H 	; wymazanie z ekranu (znak spacji)


       RET


ENDP








;***************************************************


;* Inicjalizacja myszy


;***************************************************


RESET_MOUSE PROC NEAR


       PUSH AX


       MOV  AX,0


       INT  33H     ;reset myszy


       POP  AX


ENDP





























;***************************************************


;* Pokazuje kursor myszy


;***************************************************


SHOW_MOUSE PROC NEAR


       PUSH AX


       MOV AX,1     ;pokaż kursor


       INT 33H


       POP AX


       RET


ENDP





;***************************************************


;* Schowanie kursora myszy


;***************************************************


HIDE_MOUSE PROC NEAR


       PUSH AX


       MOV AX,2    ;chowaj kursor


       INT 33H


       POP AX


       RET


ENDP








CODE ENDS


END START





Objaśnienia


W drugim z programów wykorzystano przerwanie 33H służące do obsługi myszy.


Wykorzystano następujące funkcje tego przerwania:


0  	inicjacja myszy


1 	"pokazanie" kursora myszy


2 	"chowanie" kursora myszy


3	odczyt naciśniętych przycisków i pozycji 


�
Asembler 8051


Asembler mikrokomputera jednoukładowego 8051 jest w pewnym stopniu zbliżony


do asemblera mikroprocesorów 8088/8086. 


Również i w tym przypadku aby zrozumieć prgramowanie w asemblerze dla 8051 należy zapoznać się ze strukturą rejestrów dostępnych programowo.


 Jednostka centralna mikrokomputerów serii MCS51 jest 8-bitowa. Zawiera ona następujące rejestry:


A		akumulator


B		dodatkowy rejestr stosowany przy operacjach mnożenia i dzielenia


PC		16-bitowy licznik rozkazów (ang. program counter)


DPTR	16-bitowy wskaźnik danych (ang. data pointer)


F		8-bitowy rejestr znaczników (ang. flags)


SP		8-bitowy wskaźnik stosu


4 zespoły ośmiobitowych rejestrów roboczych (po 8 rejestrów w każdym)





Ponadto mikrokomputer jednoukładowy 8051 zawiera cztery 8-bitowe układy we-wy równoległych P0, P1, P2, P3, układ we-wy szeregowych, dwa 16-bitowe układy czasowe/liczniki T0, T1 oraz układ wewnętrznego sterownika przerwań. Z układami tymi związane są dodatkowe rejestry.





TH0,TL0 	starszy i młodszy bajt układu czasowego/licznika T0


TH1,TL1	starszy i młodszy bajt układu czasowego/licznika T1


TCON	rejestr sterujący dla układów czasowych/liczników T0,T2


TMOD	rejestr sterujący trybu dla układów czasowych/liczników T0,T2





SCON	rejestr sterujący transmisją szeregową





IE		rejestr sterujący dla przerwań


IP		rejestr priorytetów dla przerwań





Bliższych informacji na temat trybów pracy powyższych układów mikrokomputerów jednoukładowych serii MCS51 oraz słów sterujących dla rejestrów można znaleźć w pozycji II ze spisu literatury.


W mikrokomputerze jednoukładowym 8051 operacje arytmetyczne mogą być wykonywane na liczbach dwójkowych i dwójkowo dziesiętnych a operacje logiczne na słowach 8-bitowych i na pojedynczych bitach. Pojedyncze bity mogą być ustawiane na 1, zerowane, negowane, przesuwane testowane i używane w operacjach logicznych. Ta ostatnia cecha znacznie upraszcza zastosowanie mikrokomputerów tej serii do kombinacyjnego bądź sekwencyjnego sterowania obiektów dyskretnych.


16-bitowy licznik rozkazów PC służy do adresowania wewnętrznej i zewnętrznej pamięci programu zapewniając przestrzeń adresowania równą 64 kB.


Wskaźnik danych DPTR jest wykorzystywany jako rejestr bazowy w rozkazach skoków pośrednich i rozkazach dotyczących operacji na tablicach (adres jest wyznaczany jako suma zawartości akumulatora i wskaźnika danych DPTR) oraz do adresowania zewnętrznej pamięci danych.


Rejestr znaczników stanu F (flags) zawiera następujące znaczniki:


CY		znacznik przeniesienia


RS1,RS0	znaczniki wyboru jednego z czterech zespołow rejestrów roboczych


OV		znacznik przepełnienia


P		znacznik parzystości


F0		znacznik programowy użytkownika ustawiany, zerowany i testowany 		programowo.


	Adresy wymienionych rejestrów rozmieszczone są w pamięci wewnętrznej procesora w sposób przedstawiony poniżej:





Nazwa�
Adres�
Opis�
�
P0  �
80h�
Port rownoległy 0�
�
SP�
81h�
Wskaźnik stosu�
�
DPTR�
82h, 83h�
Wskaźnik danych�
�
TCON�
88h�
Rejestr sterujący timerami�
�
TMOD�
89h�
Rejestr określający tryb pracy timerów�
�
TL0�
8Ah�
Młodszy bajt timera 0�
�
TL1�
8Bh�
Młodszy bajt timera 1�
�
TH0�
8Ch�
Startszy bajt timera 0�
�
TH1�
8Dh�
Starszy bajt timera 1�
�
P1�
90h�
Port równoległy 1�
�
SCON�
98h�
Rejestr sterujący portem szeregowym�
�
SBUF�
99h�
Rejestr buforowy portu szeregowego.�
�
P2�
A0h�
Port równolegly 2�
�
IE�
A8h�
Rejestr sterujący blokadą i odblokowaniem przerwań zewnętrznych.�
�
P3�
B0h�
Port równolegly 3.�
�
IP�
B8h�
Rejestr sterujący priorytetami przerwań zewnętrznych�
�
PSW�
D0h�
Rejestr znaczników�
�
ACC�
E0h�
Akumulator�
�
B�
F0h�
Rejestr B�
�






 Lista rozkazów procesora 8051





     Operacje arytmetyczne:





ADD  A,Rn       	� dodaj zawartość rejestru do ACC.


ADD  A,adr8     	� dodaj zawartość komórki pamięci do ACC.


ADD  A,@Ri      	� dodaj do ACC zawartość komórki pamięci adresowanej pośrednio.


ADD  A,#dd      	� dodaj do ACC drugi bajt rozkazu.


ADDC A,Rn       	� dodaj do ACC zawartość rejestru z przeniesieniem.


ADDC A,adr8     	� dodaj zawartość komórki pamięci do ACC  z przeniesieniem. 


ADDC A,@Ri      	� dodaj do ACC zawartość komórki pamięci z przeniesieniem.


ADDC A,#dd      	� dodaj do ACC drugi bajt rozkazu z przeniesieniem.


SUBB A,Rn       	� odejmij zawartość rejestru od ACC.


SUBB A,adr8     	� odejmij zawartość komórki pamięci od ACC.


SUBB A,@Ri      	� odejmij od ACC zawartość komórki pamięci adresowanej pośrednio.


SUBB A,#dd      	� odejmij od ACC drugi bajt rozkazu .


INC  A          	� zwiększ o 1 zawartość ACC.


INC  Rn         	� zwiększ o 1 zawartość rejestru.


INC  adr8       	� zwiększ o 1 zawartość komórki pamięci.


INC @Ri         	� zwiększ o 1 zawartość kmórki pamięci adresowanej pośrednio.


DEC  A          	� zmniejsz o 1 zawartość ACC.


DEC  Rn         	� zmniejsz o 1 zawartość rejestru.


DEC  adr8       	� zmniejsz o 1 zawartość komórki pamięci.


DEC @Ri         	� zmniejsz o 1 zawartość kmórki pamięci adresowanej pośrednio.


INC  DPTR       	� zwiększ o 1 wskaźnik danych (rejestr DPTR).


MUL AB          	� pomnóż zawartości rejestrów B i ACC.


DIV AB          	� podziel zawartość ACC przez B.


DA A            	� poprawka dziesiętna akumulatora.





     Operacje logiczne:





ANL  A,Rn       	� iloczyn logiczny zawartości rejestru i akumulatora.


ANL  A,adr8     	� iloczyn logiczny zawartości komórki  pamięci i akumulatora.


ANL  A,@Ri      	� iloczyn logiczny zawartości ACC i komórki pamięci adresowanej pośrednio.


ANL  A,#dd      	� iloczyn logiczny zawartości akumulatora z drugim bajtem rozkazu.


ANL  adr8,A     	� iloczyn logiczny zawartości ACC i komórki pamięci.


ANL  adr8,#dd   	� iloczyn logiczny zawartości komórki pamięci z drugim bajtem rozkazu.


ORL  A,Rn       	� suma logiczna zawartości rejestru i akumulatora.


ORL  A,adr8     	� suma logiczna zawartości akumulatora i komórki pamięci.


ORL  A,@Ri      	� suma logiczna zawartości ACC i komórki pamięci adresowanej pośrednio.


ORL  A,#dd      	� suma logiczna zawartości ACC z drugim bajtem rozkazu.


ORL  adr8,A     	� suma logiczna zawartości komórki pamięci i akumulatora.


ORL  adr8,#dd   	� suma logiczna zawartości komórki pamięci z drugim bajtem rozkazu.


XRL  A,Rn       	� suma exclusive�or zawartości rejestru i akumulatora.


XRL  A,adr8     	� suma exclusive�or zawartości akumulatora i komórki pamięci.


XRL  A,@Ri      	� suma exclusive�or zawartości ACC i komórki pamięci adresowanej pośrednio.


XRL A,#dd       	� suma exclusive�or  zawartości ACC z drugim bajtem rozkazu.


XRL adr8,A      	� suma exclusive�or zawartości komórki pamięci i akumulatora.


XRL adr8,#dd    	� suma exclusive�or zawartości komórki pamięci z drugim bajtem rozkazu.


CLR  A          	� zerowanie akumularora.


CPL  A          	� negacja logiczna zawartości akumulatora.


RL   A          	� rotacja zawartości akumulatora w lewo.


RLC  A          	� rotacja zawartości akumulatora w lewo wraz z bitem przeniesienia.


RR   A          	� rotacja zawartości akumulatora w prawo.


RRC  A          	� rotacja zawartościo akumulatora w prawo wraz z bitem przeniesienia.


SWAP A          	� zamiana miejscami starszego półbajtu akumulatora z młodszym.





     Operacje przesyîania danych:





MOV  A,Rn       	� przesłanie z rejestru do akumulatora.


MOV  A,adr8     	� przesłanie z pamięci do ACC.


MOV  A,@Ri      	� przesłanie z pamięci adresowanej pośrednio do akumulatora.


MOV  A,#dd      	� ładowanie akumulatora drugim bajtem rozkazu.


MOV  Rn,A       	� przesłanie z ACC do rejestru.


MOV  Rn,adr8    	� przesłanie z pamięci do rejestru.


MOV  Rn,#dd     	� ładowanie rejestru drugim bajtem rozkazu.


MOV  adr8,A     	� przesłanie z ACC do pamięci.


MOV  adr8,Rn    	� przesîanie z rejestru do pamiďci.


MOV  adr8,adr8  	� przesîanie pamięć � pamięć.


MOV  adr8,@Ri   � przesłanie z pamięci adresowanej pośrednio do pamięci adresowanej bezpośrednio.


MOV  adr8,#dd   	� ładowanie komórki pamięci drugim bajtem rozkazu.


MOV  @Ri,A      	� przesłanie z ACC do pamięci sdresowanej pośrednio.


MOV  @Ri,adr8   � przesłanie z pamięci adresowanej bezpośrednio do pamięci adresowanej pośrednio.


MOV  @Ri,#dd    � ładowanie pamięci adresowanej pośrednio drugim bajtem rozkazu.


MOV  DPTR,#dddd � ładowanie wskaźnika danych (DPTR) stałą 16 � bitową.


MOVC A,@A+DPTR  � przesłanie bajtu z pamięci (adresacja względna) do akumulatora.


MOVC A,@A+PC    � przesłanie bajtu z pamięci (adresacja względna) do akumulatora.


MOVX A,@Ri      � przesłanie z pamięci zewnętrznej adresowanej pośrednio do ACC.


MOVX A,@DPTR    � przesłanie z pamięci zewnętrznej do ACC.


MOVX @Ri,A      � przesłanie z ACC do pamięci zewnętrznej.


MOVX @DPTR,A    � przesłanie z ACC do pamięci zewnętrznej.


PUSH adr8       	� ładowanie bajtu na stos.


POP  adr8       	� pobranie bajtu ze stosu.


XCH  A,Rn       	� wymiana zawartości rejestru i ACC.


XCH  A,adr8     	� wymiana zawartości komórki pamięci wewnętrznej i ACC.


XCH  A,@Ri      	� wymiana zawartości komórki pamięci adresowanej pośrednio i ACC.


XCHD A,@Ri      � wymiana młodszych póîbajtów między pamięcią i ACC.





     Operacje na bitach :





CLR  C          	� zerowanie wskaźnika przeniesienia.


CLR  bit        	� zerowanie dowolnego bitu w rejestrze specjalnego przeznaczenia.


SETB C          	� ustawienie wskaźnika przeniesienia.


SETB bit        	� ustawienie dowolnego bitu w rejestrze specjalnego przeznaczenia.


CPL  C          	� negacja bitu przeniesienia.


CPL  bit        	� negacja dowolnego bitu w rejestrze specjalnego przeznaczenia.


ANL  C,bit      	� iloczyn logiczny bitu przeniesienia i dowolnego bitu w rejestrze specjalnego przeznaczenia


ORL  C,bit      	� suma logiczna bitu przeniesienia i dowolnego bitu w rejestrze specjalnego przeznaczenia


ORL  C,/bit     	� suma logiczna bitu przeniesienia i negacji dowolnego bitu w rejestrze specjalnego przeznaczenia


MOV  C,bit     	� przesłanie bitu z rejestru specjalnego przeznaczenia do wskaźnika przeniesienia.


MOV  bit,C      	� przesłanie bitu preniesienia na dowolny bit rejestru specjalnego przeznaczenia.





    Rozkazy skoków:





JC   ofs         	� skok względny gdy CY = 1.


JNC  ofs         	� skok względny gdy CY = 0.


JB   bit,ofs     	� skok względny gdy bit = 1.


JNB  bit,ofs     	� skok względny gdy bit = 0.


JBC  bit,ofs     	� gdy bit = 1 skok względny i zerowanie bitu.


AJMP adr11       	� skok bezwarunkowy na stronie 2kB.


LJMP adr16       	� długi skok bezwarunkowy.


SJMP ofs         	� bezwarunkowy skok względny.


JZ   ofs         	� skok względny gdy (ACC) = 0.


JNZ  ofs         	� skok względny gdy (ACC) <> 0.


CJNE A,adr8,ofs  	� skok względny gdy ACC <> [adr8].


CJNE A,#dd,ofs   	� skok względny gdy ACC <> dd.


CJNE Rn,#dd,ofs  �  skok względny gdy (Rn) <> dd.


CJNE @Ri,#dd,ofs � skok względny gdy [Ri] <> dd.


DJNZ Rn,ofs      	� zmniejsz zawartość rejestru i gdy (Rn) <> 0 �> skok względny.


DJNZ adr8,ofs    	� zmniejsz zawartość komórki pamięci i gdy [adr8] <> 0 �> skok względny.





     Rozkazy dotyczące podprogramów:





ACALL adr11     	� bezwarunkowe wywołanie podprogramu z bieżącej strony 2kB.


LCALL adr16     	� bezwarunkowe wywołanie podprogramu (pełny adres)


RET             	� bezwarunkowy powrót z podprogramu.


RETI            	� bezwarunkowy powrót z podprogramu obsługi przerwania.





     Rozkazy inne:





NOP             	� nic nie rób.





Skróty stosowane w opisie :





Rn        	� jeden z ośmiu rejestrów ( 0 <= n <= 7).


Ri        	� jeden z dwu rejestrów (r0 lub R1).


adr8     	� ośmiobitowy adres wewnętrznej pamięci danych.


@Ri       	� adres pośredni komórki wewnętrznej pamięci danych.


adr16     	� szesnastobitowy adres w pamięci programu. (używany przy skokach w przestrzeni 64KB).


adr11     	� jedenastobitowy adres w pamiďci programu. (używany przy krótkich, obejmujących 2kB skokach)


ofs       	� adres względny skoku. Liczba ze znakiem z  zakresu od �128 do 127.


bit       	� adres bitu w wewnętrznej pamięci danych.


#dd       	� stała ośmiobitowa.


#dddd     � stała szesnastobitowa.








Praca z asemblerem


Tworzenie programów dla mikrokomputera 8051 jest analogiczne jak dla procesorów 8088/86. Należy przygotować program źródłowy, a następnie poddać go asemblacji.


W ramach laboratorium do testowania napisanego programu posłużono się symulatorem mikrokomputerów serii MCS51 firmy Avsim pracującym w środowisku systemu MS DOS.





Przykład	<


Oto przykładowy program dla mikrokomputera 8051.


CPU "8051.TBL"


HOF "INT8"


;**********************************************************


;* 1. PROCEDURA MAIN czyta wzorzec bitowy z PORT-u 2


;*    i przewija go na LED'ach dołączonych do PORT-u 1


;*


;*    1 reprezentuje LED-a zaświeconego, 0 zgaszonego.


;*    Czestość zmian kieruneku przewijnia zmienia się w 


;*    zależności od wprowadzonego wzorca bitowego na 


;*    PORT-cie 2


;*


;*


;*    2. Procedura obsługi przerwania TIMER-a 0 zmienia 


;*    kierunek przewijania


;*


;*    F0      ;flaga kierunku  0 - rotacja w prawo


;*                             1 - rotacja w lewo


;*    Rejestry


;*


;*    Bank 0


;*    R2 ; wzorzec czytany z PORT-u 2


;*    R3 ; aktualny wzorzec


;***********************************************************








ORG     0





LJMP     MAIN            ; RESET








ORG     0Bh





LJMP     T0INT           ; obsługa przerwania Timera 0








ORG     100h





MAIN:


CLR A 			; zeruj akumulator


MOV P1,A    	 	; startujemy z wyłączonym LED'ami (0)


CLR F0			; poczatkowa rotacja - w prawo





; *** inicjacja parametrów i trybu Timer-a 0





MOV TMOD,#00H   ; Timer 0 tryb 0


MOV IP,#02H     ; Przerwanie timera ma największy 


                ; priorytet (miłby znaczenie, gdyby


                ; były jeszcze inne przerwania)


MOV IE,#82H 	 ; Odblokowane przerwania Timer-a 0


MOV TCON,#10H   ; Uaktywnienie Timer-a 0








NEWPTRN:MOV  R2,A   ; zachowaj wzorzec z PORT-u 2


        MOV  R3,A   ; zachowaj aktualny wzorzec


        MOV  P1,A   ; ustaw nowy wzorzec na PORT 1


        CPL  A      ; - ustaw licznik timer-a na wartość A


        MOV  TH0,A  ; /





ROTATE: MOV  A,R3         ; załaduj wzorzec


        JB   F0,ROTLEFT   ; wybierz kierunek przewijania


        RR   A            ; przewijaj w prawo


        LJMP  SETP1        ; aktualizuj PORT 1


ROTLEFT:RL   A            ; przewijaj w lewo


SETP1:  MOV  R3,A


        MOV  P1,A         ; aktualizuj PORT 1


        MOV  A,P2         ; odczyt PORTU P2





        XRL  A,R2         ; czy zmieniony wzorzec


        JZ   ROTATE       ; nie  -> kontynuuj przewijanie


        XRL  A,R2         ; tak  -> odtwórz wzorzec w A


        LJMP  NEWPTRN      ; i zacznij od nowa














;***********************************************************


;* Obsługa przerwania Timer'a 0


;***********************************************************


T0INT:  PUSH    ACC          ; schowaj rejestry


        CPL     F0           ; F0 <- !F0  (zmien kierunek)


        MOV     A,R2         


        CPL     A            


        MOV     TH0,A        ; ustaw timer


        POP     ACC          ; odtwórz rejestry


        RETI





END





Assemblery, debuggery 	� PAGE �9�	





Instytut Informatyki Politechniki Śląskiej











